# $Id: PKGBUILD 73833 2012-07-16 05:37:51Z spupykin $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>
# Contributor: Francois Charette <firmicus@gmx.net>

pkgname=perl-berkeleydb
pkgver=0.51
pkgrel=1
pkgdesc="Interface to Berkeley DB version 2, 3 or 4"
arch=('i686' 'x86_64')
url="http://search.cpan.org/dist/BerkeleyDB/"
license=('GPL' 'PerlArtistic')
depends=('perl' 'db')
options=('!emptydirs')
source=(http://search.cpan.org/CPAN/authors/id/P/PM/PMQS/BerkeleyDB-$pkgver.tar.gz)
md5sums=('50d0d18227a6568e2557a1b1b33f8f55')

build() {
  cd  $srcdir/BerkeleyDB-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  _dbver=`pacman -Q db | cut -d\  -f2 | cut -d- -f1`
  depends=('perl' "db=$_dbver")

  cd  $srcdir/BerkeleyDB-$pkgver
  make pure_install doc_install DESTDIR=$pkgdir
  find $pkgdir -name '.packlist' -delete
  find $pkgdir -name '*.pod' -delete
}
